

Estruturas de Repetição

Ranieri Alves dos Santos
ranieri.santos@sc.senai.br
SENAI/sc de Tubarão-SC

O que são?

São conjuntos de comandos que repetem instruções em uma quantidade pré-determinada ou não.

Exemplo: Verificar se um sujeito é maior ou menor de idade

```
var
idade: Inteiro

Inicio
  Escreva("Informe a sua idade")
  Leia(idade)
  se idade >= 18 entao
    Escreva("Você é Maior de Idade")
  senao
    Escreva("Você é Menor de Idade")
  FimSe
Fimalgoritmo
```

E se fossem 3 sujeitos?

O que são?

Exemplo: Verificar se três sujeitos são maiores ou menores de idade

```
Inicio
Escreva("Informe a sua idade")
Leia(idade1)
se idade1 >= 18 entao
    Escreva("Você é Maior de Idade")
senao
    Escreva("Você é Menor de Idade")
FimSe
Escreva("Informe a sua idade")
Leia(idade2)
se idade2 >= 18 entao
    Escreva("Você é Maior de Idade")
senao
    Escreva("Você é Menor de Idade")
FimSe
Escreva("Informe a sua idade")
Leia(idade3)
se idade3 >= 18 entao
    Escreva("Você é Maior de Idade")
senao
    Escreva("Você é Menor de Idade")
FimSe
Fimalgoritmo
```

Simplex?

O que são?

**As estruturas de repetição servem para estes casos.
Segue abaixo a sintaxe da estrutura de repetição com laços contados
"Para Faça":**

```
para variavel* de inicio** ate final*** faca
    //instruções para repetir
fimpara
```

* Nome da variável que irá controlar a repetição
** Valor da variável onde iniciará a repetição
*** Valor da variável onde terminará a repetição

Exemplo:

```
para i de 1 ate 3 faca
    //instruções para repetir
fimpara
```

Exemplo 1

```
var
  idade, i: Inteiro

Inicio
  para i de 1 ate 3 faca
    Escreva("Informe a sua idade")
    Leia(idade)
    se idade >= 18 entao
      Escreva("Você é Maior de Idade")
    senao
      Escreva("Você é Menor de Idade")
    fimse
  fimpara
Fimalgoritmo
```

Exemplo 2

Algoritmo para imprimir todos os números de 1 até outro número determinado pelo usuário

```
var
  n, i: Inteiro

Inicio
  Escreva("Informe o número desejado")
  Leia(n)
  para i de 1 ate n faca
    Escreva(i)
  fimpara
Fimalgoritmo
```

Exemplo 3

Algoritmo para somar os números até outro número determinado pelo usuário

```
var
  n, i, soma: Inteiro

Inicio
  Escreva("Informe o número desejado")
  Leia(n)
  soma <- 0
  para i de 1 ate n faca
    soma <- soma + i
  fimpara
  Escreva(soma)
Fimalgoritmo
```

Exemplo 4

Algoritmo para informar 10 números inteiros e ao final informar o maior deles.

```
var
  i, maior, n: Inteiro

Inicio
  maior <- 0
  para i de 1 ate 10 faca
    Escreva("Informe o número")
    Leia(n)
    se n > maior entao
      maior <- n
    fimse
  fimpara
  Escreva("O número maior é: ", maior)
Fimalgoritmo
```

Atividades

1- Escreva um algoritmo para ler o nome, o salário e a quantidade de pessoas. Ao final informe a média de salários e a pessoa que possui o maior salário.

Outras estruturas

A estrutura “**Para Faça**” repete durante a iteração de um valor. Há outras estruturas que repetem a partir de uma condição (expressão lógica).

Estrutura “Enquanto Faça”:

```
enquanto <expressão lógica> faca  
    // comandos para repetir  
fimenquanto
```

Os comandos são executados enquanto a expressão lógica for verdadeira

Enquanto Faça

Um algoritmo que soma valores informados pelo usuário até que ele deseje sair.

```
Algoritmo "Enquanto"  
var  
    op:Literal  
    num, soma: Real  
inicio  
    soma <- 0  
    op <- "s"  
    enquanto (op = "s") OU (op = "S") faça  
        Escreva("Informe o número para o cálculo")  
        Leia(num)  
        soma <- soma + num  
        Escreva("Deseja continuar? (s/n)")  
        Leia(op)  
    fimenquanto  
fimalgoritmo
```

Enquanto Faça

Um algoritmo que calcula a área de retângulos enquanto a área for menor que 200m²:

```
Algoritmo "Enquanto"  
var  
    lado1, lado2, area: Real  
inicio  
    area <- 0  
    enquanto area < 200 faça  
        Escreva("Informe o tamanho do 1° lado")  
        Leia(lado1)  
        Escreva("Informe o tamanho do 2° lado")  
        Leia(lado2)  
        area <- lado1 * lado2  
        Escreva("Área: ", area)  
    fimenquanto  
fimalgoritmo
```

Atividades

1- Escreva um algoritmo que calcule o triplo de um número informado pelo usuário enquanto o usuário não informe o zero.

Estrutura “Repita Até”

A estrutura “Repita Até” segue o mesmo princípio da “Enquanto Faça”, porém a verificação condicional é realizada no final da repetição, deste modo pelo menos uma vez a estrutura é executada.

```
repita  
    // códigos que serão repetidos  
ate <expressão lógica>
```

Repita Até

Um algoritmo que calcula a área de retângulos enquanto a área for menor que 200m²:

```
Algoritmo "repita"
var
  op:Literal
  num, soma: Real
inicio
  soma <- 0
  op <- "s"
  repita
    Escreva("Informe o número para o cálculo")
    Leia(num)
    soma <- soma + num
    Escreva("Deseja continuar? (s/n)")
    Leia(op)
  até (op = "s") OU (op = "S")
fimalgoritmo
```

Exercícios

1- Faça um algoritmo que imprima na tela as seguintes seqüências:

```
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34
35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50 51 52 53 54 55 56 57 58 59 60 61 62 63 64 65
66 67 68 69 70 71 72 73 74 75 76 77 78 79 80 81 82 83 84 85 86 87 88 89 90 91 92 93 94 95 96
97 98 99 100
```

```
2 4 6 8 10 12 14 16 18 20 22 24 26 28 30 32 34 36 38 40 42 44 46 48 50 52 54 56 58 60 62 64
66 68 70 72 74 76 78 80 82 84 86 88 90 92 94 96 98 100
```

```
60 59 58 57 56 55 54 53 52 51 50 49 48 47 46 45 44 43 42 41 40 39 38 37 36 35 34 33 32 31 30
29 28 27 26 25 24 23 22 21 20 19 18 17 16 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1 0
```

2- Faça um algoritmo que calcule a seguinte expressão:

```
1 + 2 + 3 + 4 + 5 + 6 + 7 + 8 + 9 + 10 + 11 + 12 + 13 + 14 + 15 + 16 + 17 + 18 + 19 + 20
```

3- Faça um algoritmo que informe a quantidade de compras e após isso informe para cada compra o nome do produto, o preço unitário e a quantidade de cada produto. Ao final exibir para cada compra o seguinte texto:

```
Produto: Feijão
Preço: R$ 12.00
Quantidade: 10
Total: R$ 120.00
```